
Introduzione

Il presente documento contiene una relazione del lavoro effettuato per il progetto dei filtri assegnati nel corso
di Filtri e Reti non Lineari tenuto al Politecnico di Torino, nell’anno accademico 2008/2009.

Per ciascuno dei filtri si descriveranno i passi seguiti e si proporranno i risultati calcolati, per poi proporre
il circuito finale e i risultati delle simulazioni effettuate mediante il software PSpice.

Come la precedente frase suggerisce, la relazione sarà suddivisa in 3 capitoli, ciascuno associato ad uno
dei progetti effettuati; in ciascun capitolo si cercherà di esaurire i quesiti richiesti nelle specifiche, in modo da
presentare i risultati conseguiti in seguito alle fasi di progetto e simulazione.

Le specifiche richiedono il progetto di un filtro passa alto con:

• Curva di risposta alla Chebyshev;

• Banda passante da 35 kHz a +∞, con attenuazione non superiore a αM = 1, 25 dB;

• Banda attenuata da 0 kHz a 10 kHz, con attenuazione minima αm = 55 dB.

Le tre possibili realizzazioni circuitali delle precedenti specifiche sono:

• Filtro basato su circuito LC passivo bicaricato;

• Filtro basato su circuito LC passivo monocaricato;

• Filtro basato su circuito RC attivo.

Capitolo 1

Sintesi del filtro mediante circuito LC
passivo bicaricato

In questo primo capitolo si proporranno alcuni calcoli comuni a tutte le realizzazioni possibili del filtro, calcoli
che quindi non verranno riproposti in altri capitoli del documento: il calcolo del fattore ε e quello dell’ordine
del filtro n infatti sono validi per qualsiasi realizzazione, sia essa effettuata mediante circuiti passivi o attivi.

1.1 Sintesi funzione di trasferimento passa-basso normalizzato

Dalla teoria si riprende l’espressione operativa per il calcolo del parametro ε, a partire dal quale si potrà quindi
calcolare l’ordine n del filtro:

ε =
√

100,1·αM − 1 = 0, 57751314458

Si calcola quindi, per quanto riguarda il filtro passa-alto, la pulsazione normalizzata ΩS indicante l’inizio
della banda di attenuazione, come:

ΩS =
ω0

ωS
= 3, 5 Hz

A partire da questi parametri è dunque possibile calcolare l’ordine del filtro come:

n ≥
arccosh

(√
100,1·αm − 1
|ε|

)
arccosh(ΩS)

' 4

Il filtro in questione dunque deve avere almeno ordine 4; al fine di minimizzare il numero di elementi reattivi,
quindi, si sceglie di sintetizzare un filtro di ordine 4.

Al fine di scegliere il filtro che meglio potrebbe soddisfare le specifiche, dal catalogo si sceglie quello corrispon-
dente alla sezione ”C0450”. L’idea principale è stata quella di considerare il filtro Chebyshev già sintetizzato, a
partire dunque dall’uso della riga ”T” della tabella del catalogo; mediante una verifica (in seguito proposta in
forma grafica) dell’andamento qualitativo del filtro, è stata stimata un’attenuazione minima di 54 dB, dunque
insufficiente al fine di soddisfare le specifiche di progetto.

La soluzione al problema è adottare il metodo di sintesi manuale del filtro, mediante i metodi studiati in
sede di lezione. Dalla teoria, dunque, si sa che il modulo quadro della funzione di trasferimento del circuito è
quantificabile come:

|H(jΩ)|2 =
M

1 + |h(jΩ)|2

Dal momento che si intende utilizzare una approssimazione alla Chebyshev, si può dire che:∣∣h(jΩ)2
∣∣ = ε2 |T4(Ω)|2

Dove T4(Ω) è il polinomio di Chebyshev di ordine 4; esso si può ricavare mediante sintesi a partire dalla
formula recursiva, ottenendo:

T4(jΩ) = 8Ω4 − 8Ω2 + 1

Si può dunque dire che, essendo il filtro passa-alto dotato di guadagno unitario in banda passante, M = 1;
l’espressione operativa della funzione di trasferimento nel dominio di Fourier (normalizzata in Ω) sarà dunque:

|H(jΩ)|2 =
1

1 + ε2 |8Ω4 − 8Ω2 + 1|2

Dal momento che il procedimento di sintesi viene comunemente utilizzato nel dominio della variabile s, ossia
nel dominio di Laplace, considerando l’ascissa di convergenza nulla, si può effettuare il passaggio di dominio
considerando semplicemente la seguente trasformazione:

s = jω

Quindi la funzione di trasferimento, espressa nel dominio di Laplace, avrà la seguente espressione operativa:

|H(s)|2 =
1

1 + ε2 [8s4 + 8s2 + 1]2

Il metodo di sintesi a questo punto richiede il calcolo della funzione H(s); essa si può ricavare, a partire
dalla seguente osservazione:

|H(s)|2 = H(s) ·H(−s) =
f(s) · f(−s)
g(s)g(−s)

Cosa significa ciò? Il filtro che si intende calcolare deve essere un circuito stabile, ossia tale per cui i
poli della sua funzione di trasferimento si trovino nel semipiano sinistro del dominio di Laplace (in termini di
piano complesso, nel cerchio unitario); al fine di determinare la ”corretta” H(s), sarà necessario fattorizzare,
mediante l’ausilio di un calcolatore (al fine di risparmiare tempo rispetto a metodi analitici egualmente validi se
non nell’efficienza), ricavare gli zeri del denominatore di |H(s)|2, che definiremo con la funzione g(s); dunque:

H(s) =
f(s)
g(s)

=
1
g(s)

Mediante il calcolatore, dunque, è stato possibile ricavare i seguenti valori:

s1,2,3,4 = ±0, 309643375347± j0, 403604817274

s5,6,7,8 = ±0, 128258485568± j0, 974388223702

Al fine di garantire la stabilità, come già detto, dunque, è necessario selezionare tutte le soluzioni tali da avere
poli nel semipiano sinistro del dominio s, e moltiplicarle tra di loro in modo da ”costruire” il polinomio g(s);
scrivendo solo una cifra decimale per compattezza (anche se nei calcoli successivi saranno mantenute tutte), si
descrive dunque il calcolo della funzione H(s) del filtro passa-basso normalizzato:

H(s) =
1
g(s)

=

=
1

[s− (−0, 3 + j0, 4)] · [s− (−0, 3− j0, 4)] · [s− (−0, 1 + j)] · [s− (−0, 1− j)]
Mediante il calcolatore, è stata quindi ottenuta la seguente espressione analitica di H(s) (delle quali verranno

scritte cinque cifre decimali, per quanto si continuino a conservare tutte le cifre nel calcolatore per calcoli
successivi):

H(s) =
1
g(s)

=
1

4, 62011s4 + 4, 04631s3 + 6, 39199s2 + 3, 07024s+ 1, 15478

Il fine ultimo del metodo di sintesi in uso è ricavare l’espressione operativa dell’impedenza vista in ingresso
al circuito, z(s), a partire dalle espressioni finora ottenute; si definisce il coefficiente di riflessione in ingresso al
circuito, ρ(s), come:

ρ(s) =
RG − z(s)
RG + z(s)

Questa definizione è stata introdotta al fine di poter applicare, a questo punto, il teorema di Darlington, il
quale afferma che:

|H(s)|2 + |ρ(s)|2 = 1

Esprimendo i moduli come prodotti di funzioni per le relative complesse coniugate, si ha che:

f(s) · f(−s)
g(s) · g(−s)

+ ρ(s) · ρ(−s) = 1

Quindi, ricordando che f(s) = f(−s) = 1:

|ρ(s)|2 = 1− 1
1 + |h(s)|2

−→ |ρ(s)|2 =
|h(s)|2

|g(s)|2

Considerando esclusivamente le radici negative del denominatore, si possono ricavare le due possibili espres-
sioni di ρ(s) come:

ρ(s) =
h(s)
g(s)

Da qui, finalmente, è possibile ottenere un’espressione di z(s): dal momento che si sta considerando la
sintesi di un circuito passa-basso normalizzato, RG = 1; riprendendo la definizione di coefficiente di riflessione,
e invertendola, si può ricavare z(s) come:

z(s) =
1− ρ(s)
1 + ρ(s)

=
1− h(s)

g(s)

1 + h(s)
g(s)

Finalmente:

z(s) =
g(s)− h(s)
g(s) + h(s)

A questo punto, ci si trova davanti ad una scelta, dal momento che è possibile scegliere una di due possibili
espressioni di h(s):

h(s) = ±ε
[
8s4 + 8s2 + 1

]
Si sceglie di utilizzare la versione ”negativa”, e quindi di realizzare il filtro usando:

h(s) = −ε
[
8s4 + 8s2 + 1

]
A questo punto, è possibile, con l’ausilio del calcolatore, elaborare l’espressione di z(s), mediante l’ultima

definizione ricavata; come già fatto in precedenza, si riportano solo 5 cifre decimali, continuando a mantenere
le restanti nella memoria del calcolatore:

z(s) =
2, 28362s4 + s3 + 2, 27152s2 + 0, 75878s+ 0, 42812

s3 + 0, 43790s2 + 0, 75876s+ 0, 14267

In ingresso al circuito elettronico, quindi si ”vede” un’impedenza, funzione della variabile di Laplace s, di
questo genere. Al fine di sintetizzare il circuito, quindi, si utilizza il metodo delle divisioni successive, con l’ausilio
del calcolatore elettronico. Effettuando dunque la divisione tra polinomi, si ottengono, ripetendo iterativamente
le operazioni, i seguenti risultati:

1. z1(s) = 2, 2836166941845s : si tratta di un’induttanza in serie all’ingresso;

2. y2(s) = 1, 0113623374604s : si tratta di una capacità in parallelo a z1(s);

3. z3(s) = 3, 03494304453s : si tratta di un’induttanza in serie al resto del circuito;

4. y4(s) = 0, 760990860062s+ 0, 333239313761: si tratta di una capacità in parallelo ad una conduttanza;

La resistenza di uscita si potrà calcolare come reciproco della conduttanza finale, ottenendo:

zu =
1

0, 333239313761
= 3, 00084641489

1

Quello appena ottenuto è il filtro passa-basso normalizzato in grado di soddisfare le specifiche richieste; è a
questo punto necessario effettuare, sui parametri ricavati, le seguenti operazioni:

• Trasformazione passa-basso −→ passa-alto;

• De-normalizzazione rispetto all’impedenza di ingresso del circuito.

1.1.1 Trasformazione passa-basso −→ passa-alto

Per ”convertire” il progetto attualmente realizzato, ossia un filtro passa-basso, in un filtro passa-alto, è possibile
utilizzare semplicemente dei cambi di variabile, studiati in modo da permettere una conversione del tipo di
filtro da passa-basso (considerato lo ”standard” a partire dal quale si effettuano normalmente progetti di filtri
di qualsiasi tipo) a filtri di tipo generico.

Le due trasformazioni effettuate sono sostanzialmente le seguenti:

s =
ω0

p
; Ω = −ω0

ω

Dal momento che la funzione di trasferimento del circuito è pari, il ”-” non provoca problemi; esso viene
comunque inserito per completezza.

1Non sono state sinora utilizzate unità di misura, al fine di evidenziare il fatto che si stanno esclusivamente utilizzando grandezze
normalizzate.

Al fine di studiare gli effetti della trasformazione di frequenza sulle impedenze di tipo induttivo/capacitivo,
si applica la trasformazione sulle definizioni, nel dominio di Laplace, dei due tipi di reattanze.

Per quanto riguarda le capacità, date l’ammettenza y = sc e l’ammettenza trasformata secondo la trasfor-
mazione di frequenza da passa-basso a passa alto, y:

y = sc −→ y =
ω0

p
c =

1
p · 1

ω0c

Si può quindi affermare che l’ammettenza capacitiva, in seguito alla trasformazione di frequenza, venga
trasformata in un’ammettenza induttiva di valore:

l =
1
ω0c

Per quanto riguarda le induttanza, date l’impedenza z = sl e l’impedenza trasformata secondo la trasfor-
mazione di frequenza da passa-basso a passa alto, z:

z = sl −→ z =
ω0

p
l =

1
p · 1

ω0l

Si può quindi affermare che l’ammettenza induttiva, in seguito alla trasformazione di frequenza, venga
trasformata in un’ammettenza capacitiva di valore:

c =
1
ω0l

1.1.2 De-normalizzazione rispetto impedenza

Le impedenze normalizzate, finora ricavate, rispettano la seguente regola di normalizzazione: data Z l’impe-
denza non normalizzata e z l’impedenza normalizzata rispetto all’impedenza di normalizzazione Z0, la legge di
normalizzazione è:

z =
Z

Z0
←→ Z = z · Z0

Nel circuito sono presenti resistenze, induttanze e capacità; studiando gli elementi de-normalizzati nel
dominio di Laplace, si possono trarre le seguenti conclusioni:

• Una resistenza nel dominio di Laplace si trasforma in un’altra resistenza (essendo la trasformata di Laplace
un operatore lineare, e la resistenza un coefficiente moltiplicativo rispetto alla corrente al fine di ottenere
la tensione; si può dunque dire che, de-normalizzando una resistenza, si ottiene:

R = Z0 · r

• Un’induttanza nel dominio di Laplace si trasforma nel seguente modo:

L −→ sL

De-normalizzando l’impedenza z = s · l, si ottiene:

Z = s · l · Z0

Il valore equivalente dell’induttanza, anti-trasformando nel dominio del tempo, sarà:

L = l · Z0

Ossia, il valore dell’induttanza viene moltiplicato per il valore dell’impedenza di normalizzazione;

• Una capacità nel dominio di Laplace si trasforma nel seguente modo:

C −→ 1
sC

De-normalizzando l’impedenza z = 1
sC , si ottiene dunque:

Z =
1
sC
· Z0

Il valore equivalente della capacità, anti-trasformando nel dominio del tempo, sarà:

C =
c

Z0

Ossia, il valore della capacità viene diviso per il valore dell’impedenza di normalizzazione.

1.2 Calcolo dei parametri del circuito

A partire dalle osservazioni teoriche finora introdotte è possibile calcolare i parametri finali del circuito, sem-
plicemente applicando le seguenti tre regole (differenziando i casi di impedenze resistive, capacitive, induttive),
ricavabili combinando le precedenti osservazioni:

• Per quanto riguarda il calcolo delle resistenze, si può ottenere il valore della resistenza Ri come:

Ri = ri ·R0

• Per quanto riguarda il calcolo delle induttanze, si può ottenere il valore dell’induttanza Li, a partire
dall’elemento capacitivo ci del circuito passa-basso normalizzato, come:

Li =
R0

2πf0ci

• Per quanto riguarda il calcolo della capacità, si può ottenere il valore della capacità Ci, a partire dall’ele-
mento induttivo li del circuito passa-basso normalizzato, come:

Ci =
1

2πf0liR0

A partire dai valori normalizzati precedentemente calcolati, dunque, considerando f0 = 35 kHz e R0 = 150 Ω
(seguendo le specifiche), sono stati ottenuti i seguenti valori dei parametri del circuito:

Ri = 1 · 150 Ω = 150 Ω

C1 =
1

2 · π · 35 kHz · 150 Ω · 2, 2836166941845
= 13, 2750944293 nF

L2 =
150 Ω

2 · π · 35 kHz · 1, 0113623374604
= 0, 67442951748 mH

C3 =
1

2 · π · 35 kHz · 150 Ω · 3, 0349430445259
= 9, 98873020378 nF

L4 =
150 Ω

2 · π · 35 kHz · 0, 76099086006243
= 0, 896321689324 mH

Ru = 3, 00084641489 · 150 Ω = 450, 126962234 Ω

1.3 Simulazione del comportamento in frequenza del circuito

Viene a questo punto effettuata una simulazione del circuito sintetizzato mediante i passi precedentemente
esplicati, al fine di verificarne l’effettivo comportamento in frequenza, in modo da verificare il fatto che le
specifiche siano state regolarmente soddisfatte.

L’analisi del circuito viene effettuata mediante il software PSpice, in modalità ”AC Sweep”, considerando
tre particolari: dapprima si presenterà un’analisi in banda larga, in modo da osservare l’effettivo andamento
della funzione di trasferimento del circuito; verranno quindi evidenziati alcuni dettagli, in modo da verificare
il comportamento del circuito nei punti ”critici”, ossia al termine della banda di attenuazione, e all’inizio della
banda passante.

La descrizione del circuito e la scelta delle modalità di funzionamento del simulatore viene racchiusa nel
seguente script:

V_1 1 0 AC 1
R_i 1 2 150
C_1 2 3 13.2750944293n
L_2 0 3 674.42951748u
C_3 3 4 9.98873020378n
L_4 0 4 896.321689324u
R_u 0 4 450.126962234

.AC DEC 101 8k 200k

.probe

.end

Una volta caricata la visualizzazione grafica, si visualizza la seguente uscita, mediante il comando ”Add
Trace”:

DB((V(4)/V(1))*2*SQRT(150/450.126962234))

Il comando in questione permette la visualizzazione del comportamento in frequenza del circuito, consideran-
do come uscita la tensione ai capi di Ru, in decibel, tenendo conto del fattore correttivo da applicare a circuiti
di tipo bi-caricato, come quello realizzato; infatti:

t(s) , 2
√
Ri
Ru

V4(s)
V1(s)

I risultati grafici ottenuti dal procedimento sono allegati con la presente relazione; vengono esposti alcuni
risultati significativi:

f = 10, 000 kHz −→ t(jω) = −56, 087 dB

f = 35, 078 kHz −→ t(jω) = −1, 2019 dB

Osservando inoltre che l’andamento è quello previsto per un filtro basato su approssimazione alla Chebyshev,
si può dire che il progetto risulti essere completato con successo.

1.3.1 Simulazione del comportamento in frequenza con Q(jω0) = 60

Si considera la simulazione del comportamento in frequenza del circuito considerando induttanze non ideali,
quantificando il fattore Q(jω), per la frequenza di inizio della banda passante, a 60. Dalla teoria, si sa che:

Q(jω) =
ωLs
Rs

La presenza di un Q è dunque modellizzabile mediante l’introduzione di una resistenza serie, di valore
variabile con la frequenza. Per quanto riguarda il circuito attualmente in studio, si può ricavare che:

Rs =
ωLs
Q(jω)

L2 −→ Rs2 = 2, 47191328761Ω

L4 −→ Rs4 = 3, 28519057423Ω

Lo script PSpice utilizzato per la simulazione è quindi il seguente:

V_1 1 0 AC 1
R_i 1 2 150
C_1 2 3 13.2750944293n
L_2 0 5 674.42951748u
R_2 5 3 2.47191328706
C_3 5 4 9.98873020378n
L_4 0 6 896.321689324u
R_4 6 4 3.28519057351
R_u 0 4 450.126962234

.AC DEC 101 9k 500k

.probe

.end

Considerando induttori non ideali, si nota una variazione della funzione di trasferimento per frequenze
minori di quella di taglio. Questo fenomeno è dovuto al fatto che, induttori non ideali sono caratterizzati da
una resistenza serie, che per frequenze elevate non viene presa in considerazione a causa del comportamento
dellinduttanza che simula un circuito aperto.

1.3.2 Simulazione del comportamento in frequenza con aumento dei componenti
del 20% dal valore nominale

Seconda simulazione riguardante i comportamenti ”anomali” del circuito coinvolge l’aumento di tutti i parametri
reattivi del 20 % rispetto al valore nominale. Viene immediatamente presentato lo script PSpice, contenente i
valori originali, moltiplicati per un fattore 1,2 (aggiungendo dunque il 20 % rispetto al valore originale).

V_1 1 0 AC 1
R_i 1 2 150
C_1 2 3 15.93011331516n
L_2 0 3 809.315420976u
C_3 3 4 11.986476244536n
L_4 0 4 1.0755860271888m
R_u 0 4 450.126962234

.AC DEC 101 9k 500k

.probe

.end

Si nota una traslazione della funzione di trasferimento a frequenza più alte, quindi un aumento della frequenza
di taglio.

Capitolo 2

Sintesi del filtro mediante circuito LC
passivo monocaricato

Essendo il progetto del tutto analogo a quello precedentemente affrontato, si può anticipare il fatto che una
soluzione ”esatta” può essere ricavata esclusivamente mediante sintesi manuale, dunque non mediante l’uso di
tavole e cataloghi. La sintesi di un filtro monocaricato tuttavia esula dalle competenze attese dal progettista
nel presente corso, di conseguenza l’uso del manuale, in una situazione come la presente, è obbligatorio.

Si anticipa un ulteriore fatto: il catalogo, al fine di sintetizzare filtri basati su approssimazioni alla But-
terworth o alla Chebyshev (quest’ultimo nella fattispecie è il caso in studio), modifica il circuito filtrante per
renderlo ”simile” ad un filtro ellittico (”alla Cauer”), dunque l’andamento del filtro sintetizzato non sarà esat-
tamente quello di un filtro Chebyshev, ma comunque abbastanza simile (come si discuterà, confrontando con i
risultati precedentemente ottenuti).

Le disquisizioni teoriche affrontate nel precedente capitolo non verranno riprese, dunque si proporranno
immediatamente i metodi di progetto utilizzati ed il calcolo dei parametri del filtro.

Scegliendo dal catalogo la sezione ”C0450c”, riga ”T” (riferita alla sintesi mediante polinomio di Chebyshev),
si seleziona la seconda colonna, riferita al circuito ”A”, considerando r1 =∞, r2 = 1:

Sfruttando il teorema di reciprocità, applicabile su di una rete passiva, si sceglie di introdurre il circuito sui
morsetti ”a destra”, e di prelevare l’uscita ”a sinistra”, ossia di ”girare” il circuito.

L’impedenza di uscita del circuito, in questo modo, è infinita, dunque il circuito è, come le specifiche
richiedono, monocaricato; applicando la trasformazione di frequenza, combinata con la de-normalizzazione per
l’impedenza Z0, è possibile ricavare i seguenti valori (utilizzando una teoria del tutto analoga alla precedente):

Ri = 1 · 150 Ω = 150 Ω

C1 =
1

2 · π · 35 kHz · 150 Ω · 0, 835897
= 36, 2667018252 nF

L2 =
150 Ω

2 · π · 35 kHz · 1, 609548
= 0, 423778982206 mH

C3 =
1

2 · π · 35 kHz · 150 Ω · 0, 835897
= 19, 5629819756 nF

L4 =
150 Ω

2 · π · 35 kHz · 1, 614017
= 0, 422605594149 mH

2.1 Simulazione del comportamento in frequenza del circuito

Analogamente a quanto fatto per quanto concerne il primo progetto, viene ora presentata una simulazione
PSpice atta a evidenziare i risultati ottenuti in seguito alla fase di progetto.

La descrizione del circuito e la scelta delle modalità di funzionamento del simulatore viene racchiusa nel
seguente script:

V_1 1 0 DC 0V AC 1
R_i 1 2 150
C_1 2 3 36.2667018252n
L_2 0 3 0.423778982206m
C_3 3 4 19.562981976n
L_4 0 4 0.422605594149m

.AC DEC 101 8k 200k

.probe

.end

Una volta caricata la visualizzazione grafica, si visualizza la seguente uscita, mediante il comando ”Add
Trace”:

DB(V(4)/V(1))

Il comando in questione permette la visualizzazione del comportamento in frequenza del circuito, consideran-
do come uscita la tensione ai capi di Ru, in decibel; in ambito di filtri mono-caricati, non è necessario introdurre
correzioni legate al rapporto delle resistenze di ingresso e uscita, quindi l’espressione utilizzata è più semplice
della precedente.

I risultati grafici ottenuti dal procedimento sono allegati con la presente relazione; vengono esposti alcuni
risultati significativi:

f = 10, 000 kHz −→ t(jω) = −53, 457 dB

f = 35, 055 kHz −→ t(jω) = −1, 2251 dB

Come previsto teoricamente all’inizio della fase di progetto, il circuito non è in grado di soddisfare la
condizione di attenuazione minima in banda attenuata; senza ricorrere alla sintesi manuale, tuttavia, questo è
il miglior risultato ottenibile.

2.1.1 Simulazione del comportamento in frequenza con Q(jω0) = 60

Analogamente a quanto precedentemente fatto, si considera la simulazione del comportamento in frequenza
del circuito considerando induttanze non ideali, quantificando il fattore Q(jω), per la frequenza di inizio della
banda passante, a 60. Il procedimento teorico utilizzato è del tutto analogo a quello precedentemente presentato,
dunque si sceglie di presentare immediatamente lo script PSpice utilizzato per la simulazione:

V_1 1 0 DC 0V AC 1
R_i 1 2 150
C_1 2 3 36.2667018252n
L_2 0 5 0.423778982206m
R_2 5 3 1.55323109345
C_3 3 4 19.562981976n
L_4 0 6 0.422605594149m
R_4 6 4 1.54893040142

.AC DEC 101 9k 500k

.probe

.end

Valutando induttori reali con Q diverso da zero, si ottiene un peggioramento delle caratteristiche del fil-
tro uscendo fuori dalle specifiche di progetto. Alla frequenza di taglio di banda passante si ha un aumento
dellattenuazione.

2.1.2 Simulazione del comportamento in frequenza con aumento dei componenti
del 20% dal valore nominale

Seconda simulazione riguardante i comportamenti ”anomali” del circuito coinvolge l’aumento di tutti i parametri
reattivi del 20 % rispetto al valore nominale. Viene immediatamente presentato lo script PSpice, contenente i
valori originali, moltiplicati per un fattore 1,2 (aggiungendo dunque il 20 % rispetto al valore originale).

V_1 1 0 DC 0V AC 1
R_i 1 2 150
C_1 2 3 43.52004219024n
L_2 0 3 0.5085347786472m
C_3 3 4 23.4755783712n
L_4 0 4 0.5071267129788m

.AC DEC 101 9k 500k

.probe

.end

Si modificano i valori dei componenti LC del filtro aumentandoli del 20 %. Si nota una riduzione della banda
attenuata, quindi una frequenza di taglio minore rispetto alle specifiche.

Capitolo 3

Sintesi del filtro mediante circuito RC
attivo

La terza realizzazione del filtro in questione è effettuata mediante circuiti basati su componenti attivi, nella
fattispecie su amplificatori operazionali reazionati mediante reti RC passivi

L’idea alla base del progetto è suddivisa nelle seguenti fasi:

• Caratterizzazione di una singola cella RC attiva biquadratica (ossia in grado di realizzare circuitalmente
una funzione di trasferimento di ordine 2), ricavando la funzione di trasferimento;

• Unione di più celle, inserite in cascata tra loro, al fine di ottenere il filtro.

Una cella RC attiva biquadratica, come già accennato, è in grado di realizzare solo un particolare tipo di
filtri: quelli di ordine due. Dal momento che le specifiche di progetto richiedono, come calcolato nel primo
capitolo, un filtro di ordine quattro, una sola cella non sarà sufficiente. Dal momento che il filtro è di ordine
pari, tuttavia, osservazione immediatamente effettuabile è la seguente: è sufficiente un solo tipo di celle, dal
momento che un filtro passa-alto di quarto ordine è realizzabile mediante una cascata di due celle di tipo anche
uguale, purchè siano entrambe di tipo passa-alto, con parametri ovviamente differenti.

Fattorizzazione della funzione di trasferimento

H(s) =
Ks4

as4 + bs3 + cs2 + ds+ e
=

K1s
2

a1s2 + b1s+ c1
· K2s

2

a2s2 + b2s+ c2

La funzione di trasferimento del filtro, H(s), si ricava a partire dalla funzione HLP (s), ossia dalla funzione
passa-basso normalizzata precedentemente ricavata, mediante il procedimento di sintesi manuale:

HLP (s) =
1

4, 62011s4 + 4, 04631s3 + 6, 39199s2 + 3, 07024s+ 1, 15478
Sostituendo, mediante il calcolatore, alla variabile s l’espressione:

s −→ ω0

p

Si ottiene la seguente espressione, presentata in forma approssimata (nonostante sul calcolatore si continuino
a mantenere tutte le cifre significative):

H(p) =
0, 866p4

(p2 + 526, 28 · 103p+ 186, 88 · 109) · (p2 + 58, 4 · 103p+ 50 · 109)

Si propongono ora, scritti con tutte le cifre che il calcolatore propone, i coefficienti della funzione di
trasferimento ora proposta:

K = 0, 86596432336

a1 = 1; b1 = +526, 278861631 · 103; c1 = 186, 883969746 · 109

a2 = 1; b2 = +58, 403604481 · 103; c2 = 50, 0692931847 · 109

La fattorizzazione è quindi ultimata: si dispone dei parametri che si intende introdurre in ciascuna cella.

Calcolo della funzione di trasferimento della cella

Mediante l’uso del metodo dei nodi, è stato possibile ricavare l’espressione della funzione di trasferimento della
cella proposta per il progetto. (V1 − Vi)sC1 + (V1 − Vu)G2 + (V1 − V2)sC3 = 0

(V2 − V1)sC3 + V2G4 = 0
(V3 − Vu)G6 + V3G5 = 0

Si noti che:

Vu = (V + − V −) ·A

Vu = (V2 − V3) ·A = V2A− V3A

Da qui:

V3 =
V2A− Vu

A

Una volta ricavata V3, si può sostituire nella terza relazione, ricavando:(
V2A− Vu

A
− Vu

)
G6 +

V2A− Vu
A

G5 =
(
V2A− Vu − VuA

A

)
G6 +

V2A− Vu
A

G5 = 0 =

=
V2AG6

A
+
V2AG5

A
− VuG6

A
− VuG5

A
= 0

Raccogliendo:

V2(G6 +G5) = Vu

(
G6

A
+G6 +

G5

A

)
Quindi:

V2 =
Vu

G6 +G5

(
G6 +AG6 +G5

A

)
A questo punto è possibile sostituire questa espressione nella seconda equazione:[

Vu
G6 +G5

(
G6 +AG6 +G5

A

)
− V1

]
sC3 +

Vu
G6 +G5

(
G6 +AG6 +G5

A

)
G4 = 0

Si riordina a questo punto l’espressione, ottenendo:

VusC3

G6 +G5

(
G6 +AG6 +G5

A

)
− V1sC3 +

Vu
G6 +G5

(
G6 +AG6 +G5

A

)
G4 = 0

Si ricava quindi V1:

V1 =
Vu

sC3(G6 +G5)

[
sC3

(
G6 +AG6 +G5

A

)
+G4

(
G6 +AG6 +G5

A

)]
Modificando a questo punto la prima equazione, si ottiene:

V1(sC1 +G2 + sC3)− V2sC3 − VisC1 − VuG2 = 0

Sostituendo l’espressione di V1 ricavata:

Vu
sC3(G6 +G5)

[
sC3

(
G6 +AG6 +G5

A

)
+G4

(
G6 +AG6 +G5

A

)]
(sC1 +G2 + sC3)−

+
Vu

G6 +G5

(
G6 +AG6 +G5

A

)
sC3 − VisC1 − VuG2 = 0

Quindi:

Vu

{[
sC3(G6 +AG6 +G5)
sC3(G6 +G5)A

+
G4(G6 +AG6 +G5

sC3(G6 +G5)A

]
(sC1 +G2 + sC3)− G4(G6 +AG6 +G5

(G6 +G5)A
sC3 −G2

}
= VisC1

Semplificando:

Vu

{[
G6 +AG6 +G5

(G6 +G5)A
+
G4(G6 +AG6 +G5

sC3(G6 +G5)A

]
(sC1 +G2 + sC3)− G4(G6 +AG6 +G5

(G6 +G5)A
sC3 −G2

}
= VisC1

Si raccoglie quindi il contenuto della parentesi in modo da raggruppare, nel seguente modo:

Vu

{
[sC3(G6 +AG6 +G5) +G4(G6 +AG6 +G5)] (sC1 +G2 + sC3)− ...

A(G5 +G6)sC3

...+ (G6 +AG6 +G5)(sC3)2 −G2 [A(G5 +G6)sC3]
A(G5 +G6)sC3

}
= Vi(sC1)

Si può a questo punto scrivere, invertendo l’equazione, la seguente relazione:

Vu =
VisC1 [A(G5 +G6)sC3]

[sC3(G6 +AG6 +G5) +G4(G6 +AG6 +G5)] (sC1 +G2 + sC3)− (G6 +AG6 +G5)(sC3)2 −G2 [A(G5 +G6)sC3]

=
VisC1 [A(G5 +G6)sC3]

(s2C1C3)(G6 +AG6 +G5) + sC3(G6 +AG6 +G5)G2 + s2C2
3 (G6 +AG6 +G5) +G4(G6 +AG6 +G5)sC1 + ...

VisC1 [A(G5 +G6)sC3]
...+G4(G6 +AG6 +G5)G2 +G4(G6 +AG6 +G5)sC3 − (G6 +AG6 +G5)(sC3)2 −G2A(G5 +G6)sC3

=
VisC1 [A(G5 +G6)sC3]

(s2C1C3)(G6 +AG6 +G5) + s {C3(G6 +AG6 +G5)G2 + C1(G6 +AG6 +G5)G4 + C3(G6 +AG6 +G5)G4 − ...

VisC1 [A(G5 +G6)sC3]
...+ C3(G5 +G6)AG2}+G4(G6 +AG6 +G5)G2

=
VisC1 [A(G5 +G6)sC3]

(s2C1C3)(G6 +AG6 +G5) + s {(G6 +AG6 +G5)(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2}+ ...

VisC1 [A(G5 +G6)sC3]
...+G4(G6 +AG6 +G5)G2

=
VisC1 [A(G5 +G6)sC3]

s2 + s

{
(G6 +AG6 +G5)(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2

C1C3(G6 +AG6 +G5)

}
+
G4(G6 +AG6 +G5)G2

C1C3(G6 +AG6 +G5)

Quindi:

Vu
Vi

=

s2[A(G5 +G6)]
G6 +AG6 +G5

s2 + s

{
(G6 +AG6 +G5)(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2

C1C3(G6 +AG6 +G5)

}
+
G4(G6 +AG6 +G5)G2

C1C3(G6 +AG6 +G5)

Giunti a questa espressione, è possibile calcolare i parametri della cella, ossia la pulsazione ω0 e il fattore di
qualità qp:

ω2
0 =

G2G4

C1C3
−→ ω0 =

√
G2G4

C1C3

Si può dunque ottenere:

qp =

√
G2G4

C1C3

(G6 +AG6 +G5)(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2

(C1C3)(G6 +AG6 +G5)

=

=

√
G2G4

C1C3
(C1C3)(G6 +AG6 +G5)

(G6 +AG6 +G5)(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2
=

=
√
G2G4C1C3(G6 +AG6 +G5)

(G6 +AG6 +G5)(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2
=

=
√
G2G4C1C3(G6 +AG6 +G5)

(G6 +AG6 +G5)
[
(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2

G6 +AG6 +G5

] =

=
√
G2G4C1C3[

(C3G2 + C1G4 + C3G4)− C3(G5 +G6)AG2

G6 +AG6 +G5

] =

=

√
G2G4C1C3 ·

1
C3G4

C3G2

C3G4
+
C1G4

C3G4
+
C3G4

C3G4
− C3(G5 +G6)AG2

C3G4(G6 +AG6 +G5)

Riordinando quest’ultima espressione, è possibile ottenere un’espressione operativa del fattore di qualità qp:

qp =

√
C1G2

C3G4

1 +
C1

C3
+
G2

G4

[
1− A(G5 +G6)

G6 +AG6 +G5

]
Si decide, al fine di preparare il calcolo del prodotto guadagno-sensibilità, di introdurre una semplificazione

dell’espressione di qp, mediante l’uso della seguente sostituzione:

µ =
A

1 +
A

µi

Dove µi è il guadagno della cella:

µi = 1 +
R6

R5
= 1 +

G5

G6
=
G5 +G6

G6

Si sostituisce quindi l’espressione di µi in µ:

µ =
A

1 +
A

G5 +G6

G6

=
A

1 +
AG6

G5 +G6

=
A(G5 +G6)

G5 +G6 +AG6

Si noti a questo punto il fatto che è possibile ricondurre questa espressione ad una contenuta in qp; se infatti
si considera - (µ− 1), si ottiene ciò:

−(µ− 1) = −
(

A(G5 +G6)
G6 +AG6 +G5

− 1
)

= −A(G5 +G6)− (G6 +AG6 +G5)
G6 +AG6 +G5

=
−AG5 +G5 +G6

G6 +AG6 +G5

Sostituendo questa espressione in quella di qp, si ottiene la seguente semplificazione:

qp =

√
C1G2

C3G4

1 +
C1

C3
− G2

G4
(µ− 1)

18



3.0.3 Calcolo del prodotto guadagno-sensibilità

Al fine di ottimizzare il progetto del filtro, è necessario calcolare a questo punto un’espressione operativa del
prodotto guadagno-sensibilità, dove la sensibilità è relativa alle variazioni del fattore di qualità qp al variare
di A. Idealizzando l’amplificatore operazionale utilizzato per la realizzazione di ciascuna cella, considerandolo
dunque con A→∞, si intende calcolare:

ΓA = lim
A→+∞

A · Sqp

A

Dunque:

Sqp

A = Sqp
µ · S

µ
A =


S

√
C1G2

C3G4

1 +
C1

C3
− G2

G4
(µ− 1)

µ


·

S

A

1 +
A

µi
A



Sqp
µ = S

√√√√C1G2

C3G4
µ − S

1+
C1

C3
−
G2

G4
(µ−1)

µ =

= −
(−µ)

G2

G4
S
−(µ−1)

G2

G4
µ

1 +
C1

C3
− (µ− 1)

G2

G4

=
µ
G2

G4
S
−µ
G2

G4
µ

1 +
C1

C3
− (µ− 1)

G2

G4

=

=
µ
G2

G4

1 +
C1

C3
− (µ− 1)

G2

G4

Per quanto riguarda SµA:

SµA = S

A

1 +
A

µ
A = SAA − S

1+
A

µi
A = 1−

1 · S1
A +

A

µi
· S

A

µi
A

1 +
A

µi

=

= 1−

A

µi

1 +
A

µi

= 1− A

µi +A
=

µi
µi +A

=
1

1 +
A

µi

·Sqp

A = Sqp
µ · S

µ
A =

µ
G2

G4

1 +
C1

C3
− (µ− 1)

G2

G4

· 1

1 +
A

µi

A questo punto è possibile calcolare il prodotto guadagno-sensibilità:

lim
A→+∞

A · Sqp

A

Si noti innanzitutto che, per A→ +∞, µ −→ µi:

lim
A→+∞

µ =
A

1 +
A

µi

= µi

Quindi:

ΓA = lim
A→+∞

A · Sqp

A = lim
A→+∞

µ2
i

G2

G4

1 +
C1

C3
− µi

G2

G4
+
G2

G4

=
G2

(
1 +

G5

G6

)2

1 +
C1

C3
−
(

1 +
G5

G6

)
G2

G4
+
G2

G4

=

=
G2

(
1 +

G5

G6

)2

1 +
C1

C3
− G5

G6

G2

G4

Al fine di minimizzare il prodotto guadagno-sensibilità ΓA, quindi, è necessario soddisfare la seguente
condizione:

C1 � C3

3.0.4 Formule di progetto

Al fine di completare il discorso teorico, si propone la dimostrazione delle formule di progetto in seguito utilizzate
in combinazione con il software MATLab; le due espressioni fondamentali riguardano il calcolo di due parametri
del circuito: R2 e R6

Calcolo di R2

R4 = PR2; fp =
1

2pi
√
C1R2C3R4

Sostituendo la prima nella seconda, si ottiene:

R2 =
1

2πfp
√
C1C3PR4

Calcolo di R6

Partendo dall’espressione operativa di qp:

qp =

√
R4C1

R2C3

1 +
C1

C3
− R4R6

R2R5

=

√
R4C1

R2C3

R2R5C3 + C1R2R5 − C3R4R6

C3R2R5

L’espressione si può invertire nel seguente modo:

R2R5C3 + C1R2R5 − C3R4R6 =

√
R4C1

R2C3
C3R2R5

qp

Da qui, si può ricavare R6 come:

R6 =
R2R5C3

R4C3
+
C1R2R5

R4C3
−

√
R4C1

R2C3
C3R2R5

qpR4C3
=

= R5

R2

R4
+
C1R2

R4C3
−

√
R4C1

R2C3
R2R5

qpR4

 =

= R5

[
1
P

(
1 +

C1

C3

)
− 1
Pqp

√
R4C1

R2C3

]
Ordinando, si può ricavare la seguente espressione:

R6 = R5

[
1
P

(
1 +

C1

C3

)
− 1
qp

√
C1

PC3

]

3.1 Calcolo dei parametri delle celle

Il filtro RC, come già accennato, viene sintetizzato mediante l’uso di due celle RC biquadratiche basate sull’uso
di un amplificatore operazionale; al fine di riprodurre, mediante le suddette celle, la funzione di trasferimento
precedentemente sintetizzata, è stato necessario percorrere un certo numero di passi, che ora verranno descritti:

• A partire dalla funzione di trasferimento fattorizzata in termini con denominatore polinomiale di secondo
grado, e numeratore di secondo grado monomiale (precedentemente proposta), si seleziona l’ordine delle
celle da inserire in cascata, in modo da avere celle in cascata con qp crescente.

• Calcolo dei guadagni Ki per ciascuna i-esima cella, secondo i seguenti criteri:

– Il massimo della prima cella, T1,max, moltiplicato per il K1, della prima cella, deve essere pari al
massimo della funzione di trasferimento finale (sintetizzata precedentemente);

– Per la i-esima cella, il prodotto Ti,max·Ki, moltiplicato per tutti i Tj,max·Kj delle celle per j ∈ [1; i−1],
deve essere pari al massimo della funzione di trasferimento finale.

Avendo nel progetto solo due celle, e guadagno massimo in banda passante pari a 1 (0 dB) il criterio per
il calcolo dei Ki si riduce a:

T1,max ·K1 = 1

Poichè T1,max ' qp,1, si può dire che:

qp,1 ·K1 = 1 −→ K1 =
1
qp,1

Per quanto riguarda il calcolo di K2, si parte dalla seguente relazione:

K1T1,max ·K2T2,max = 1

Il calcolo di T2,max è stato effettuato mediante il seguente script MATLab:

% Progetto di un filtro RC attivo con celle in cascata
% Calcolo del massimo di |T2|
% Prima cella
wr1 = sqrt(186.883969746*10^9);
qr1 = wr1/(526.278861631*10^3);
T1_Max = qr1
% Seconda cella
wp2 = sqrt(50.0692931847*10^9);
qp2 = wp2/(58.403604481*10^3);
%
%
f = linspace(2000, 50000, 14010);
w = 2*pi * f;
p = 0 + w*j;
%
% Evaluate partial transfer function T2 on the j-omega axis:
nT2a = p.^2;
nT2=nT2a.*(p.^2);
dT2a = (p.^2 + p.*(wr1./qr1) + (wr1).*(wr1));
dT2 = dT2a.*(p.^2 + p.*(wp2./qp2) + (wp2).*(wp2));
T2 = nT2./dT2;
%
% Evaluate amplification:
ampl = abs(T2);
T2_Max=abs(max(T2))

%
% Plot attenuation:
plot(f, ampl);
grid;
K_tot=0.86596432336;
K_2=T1_Max/T2_Max
K_1=K_tot/K_2

In realtà, per semplicità, il calcolo dei Ki è incluso nello script; quella appena proposta è il procedimento
teorico seguito.

In seguito ai calcoli, effettuati dallo script, sono stati ricavati i seguenti valori:

K1 = 1, 21739047475583

K2 = 0, 71132832178081

Una volta calcolati questi parametri ”globali”, si lavora sulle singole celle, in modo da riprodurre la funzione
di trasferimento finale.

3.1.1 Cella 1

La prima cella del filtro, secondo il criterio dei qp crescenti, realizza la seguente funzione di trasferimento (dato
qp calcolato a priori):

H(p) =
1, 21739047475583p2

p2 + 58, 4 · 103p+ 50 · 109

Si possono semplicemente calcolare, a partire da questa funzione di trasferimento, i seguenti parametri:

ωp,1 = 4, 32300786196370432 · 105 rad/s

qp,1 = 0, 82142912762375

A questo punto è necessario dimensionare i parametri della cella, in modo da ottenere i suddetti valori;
al fine di minimizzare il prodotto guadagno-sensibilità ΓA, è stato scelto C1 � C3; a partire dalle formule di
progetto precedentemente ricavate, è stato utilizzato il seguente procedimento:

• Sono stati scelti i valori delle capacità, in modo da minimizzare ΓA;

• Seguendo lo schema a blocchi allegato con le specifiche, le cui formule sono state precedentemente ricavate,
sono stati ricavati i restanti parametri del circuito;

Mediante uno script MATLab, sono stati dunque ricavati i seguenti valori:

C1,1 = 100 nF

C1,2 = 0 nF

C3 = 1 nF

R2 = 31, 47872218445133 Ω

R4 = 1, 699850997960372 kΩ

R6 = 2, 137117626872152 kΩ

R5 = 10 kΩ

Utilizzando questi valori, il K effettivo della cella risulta essere il seguente:

K = 1, 21371176268722

Con un prodotto guadagno-sensibilità pari a:

ΓA = 0, 889

Lo script MATLab utilizzato è il seguente:

%%%CELLA 1%%%

wp1 = sqrt(186.883969746*10^9)
qp1 = wp1/(526.278861631*10^3)
fp1=wp1/(2*pi)
C11=100e-9
C12=22e-9
C3=0.1e-9
C1=C11+C12;

P=((C1/C3)/(4*(qp1^2)))*(sqrt(1+12*qp1^2*(1+(C3/C1)))-1)^2

R2=1/(2*pi*fp1*sqrt(P*C1*C3))
R4=P*R2
R5=10000
R6=R5*((1/P)*(1+(C1/C3))-(sqrt(C1/(P*C3))*(1/qp1)))
K=(C11/C1)*(1+(R6/R5))
GSP=qp1*(1+R6/R5)^2*sqrt((P*C3)/C1)

%%%CELLA 2%%%
wp2 = sqrt(50.0692931847*10^9)
qp2 = wp2/(58.403604481*10^3)
fp2=wp2/(2*pi)
C11=100e-9
C12=22e-9
C3=0.1e-9
C1=C11+C12;

P=((C1/C3)/(4*(qp2^2)))*(sqrt(1+12*qp2^2*(1+(C3/C1)))-1)^2

R2=1/(2*pi*fp2*sqrt(P*C1*C3))
R4=P*R2
R5=10000
R6=R5*((1/P)*(1+(C1/C3))-(sqrt(C1/(P*C3))*(1/qp2)))
K=(C11/C1)*(1+(R6/R5))
GSP=qp2*(1+R6/R5)^2*sqrt((P*C3)/C1)

p=0+j*w;
T1=((p.*p))/(((p.*p)-((567.67*10^3)*p)+(186.88*10^9)))
x=max(T1)

Seguendo il flow-chart allegato, è stato calcolato un primo valore di P , definito come:

P =
R4

R2

La prima iterazione ha portato a una R6 negativa; come si può osservare dall’espressione operativa di R6:

R6 = R5

[
1
P

(
1 +

C1

C3

)
−
√

C1

PC3

1
qp

]
Al fine di rendere positiva R6, è necessario ridurre P , in modo da rendere prevalente il termine positivo della

somma.
Dopo varie iterazioni, si imposta P = 54, quindi si ricavano i valori precedentemente esposti.

3.1.2 Cella 2

La seconda cella del filtro, secondo il criterio dei qp crescenti, realizza la seguente funzione di trasferimento (dato
qp calcolato a priori):

H(p) =
0, 71132832178081p2

p2 + 526, 28 · 103p+ 186, 88 · 109

Si possono semplicemente calcolare, a partire da questa funzione di trasferimento, i seguenti parametri:

ωp,2 = 2, 23761688375602 · 105 rad/s

qp,2 = 3, 83129928989908

Seguendo lo stesso procedimento precedentemente esposto, sono stati ricavati i seguenti valori:

C1,1 = 68 nF

C1,2 = 47 nF

C3 = 1 nF

R2 = 24, 17934796217750 Ω

R4 = 71, 82673310105153 kΩ

R6 = 2, 250676509173192 kΩ

R5 = 10 kΩ

Utilizzando questi valori, il K effettivo della cella risulta essere il seguente:

K = 0, 7243878283685

Con un prodotto guadagno-sensibilità pari a:

ΓA = 9, 241

In questo caso, è stato sufficiente utilizzare il P legato alla formula presente sul flow-chart; dai calcoli è stato
possibile ricavare:

P = 2, 97 · 103

3.2 Simulazione del comportamento in frequenza del circuito

Al fine di effettuare un’analisi in frequenza del comportamento del circuito, viene introdotto un modello ideale
di amplificatore operazionale, mediante l’opzione SUBCKT di PSpice; si è scelto di realizzare un amplificatore
modellizzato sostanzialmente con un generatore di tensione pilotato in tensione, con un guadagno pari a 108; lo
script PSpice utilizzato per la simulazione è il seguente:

* Definizione amplificatore operazionale *

.SUBCKT OA 1 2 3
* | | |
* | | | output
* | inverting input
* non inverting input
E1 3 0 1 2 1E8
.ENDS OA

* Cella 1 (CA) *

.SUBCKT CA 1 5
C_11 2 1 100n
C_3 3 2 1n
R_2 2 5 31.47872218445133
R_4 0 3 1.699850997960372k
R_5 0 4 10k
R_6 4 5 2.137117626872152k
XOA 3 4 5 OA
.ENDS CA

* Cella 2 (CB) *

.SUBCKT CB 1 5
C_11 2 1 68n
C_12 2 0 47n
C_3 3 2 0.1n
R_2 2 5 24.1793479621775
R_4 0 3 71.82673310105153k
R_5 0 4 10k
R_6 4 5 2.250676509173192k
XOA 3 4 5 OA
.ENDS CB

* Definizione circuito *

V1 1 0 AC 1
XCA 1 2 CA
XCB 2 3 CB

* Opzioni di analisi *

.AC DEC 101 9k 500k

.probe

.end

Dalle simulazioni effettuate, si può osservare il fatto che il progetto del filtro sia andato a buon fine: le
specifiche, sia per quanto riguarda la banda attenuata sia per quanto riguarda le attenuazioni minima e massima
in banda passante si possono dire rispettate.

3.2.1 Simulazione del comportamento in frequenza con ft = 1, 5 MHz

Al fine di introdurre un modello di amplificatore operazionale con una frequenza di taglio ft non infinita, si
realizza la seguente idea: utilizzando un generatore di corrente pilotato in tensione, con transconduttanza gm, si
alimentano una resistenza R e una capacità C ideali; al variare della frequenza, varierà la resistenza equivalente
della capacità, dunque la tensione che cadrà su di essa; prelevando questa tensione con un generatore di tensione
pilotato in tensione con guadagno unitario, infine, si disaccoppia l’impedenza di uscita del modello con eventuali
carichi.

Il calcolo della pulsazione di taglio ωt si basa sul seguente fatto:

ωt ' A0 · ωr
Dove:

A0 = gm ·R;ωr =
1
RC

Si può dunque scrivere che:

2πft ' gm ·R
1
RC
−→ C =

gm
2πft

Si è dimensionato un A0 = 107, con gm = 1 · 103, R = 10 kΩ; in tali condizioni, C = 106, 103295395µF.
La simulazione del comportamento in frequenza del circuito è stata dunque realizzata con il seguente script

PSpice:

* Definizione amplificatore operazionale *

.SUBCKT OAI 1 2 3
* | | |
* | | | output
* | inverting input
* non inverting input
E1 3 0 1 2 1E8
.ENDS OAI

.SUBCKT OA 1 2 3
* | | |
* | | | output
* | inverting input

* non inverting input
G_1 0 4 1 2 1000
R_2 4 0 10k
C_3 4 0 106.103295395u
E_1 3 0 4 0 1
.ENDS OA

* Cella 1 (CA) *

.SUBCKT CA 1 5
C_11 2 1 100n
C_3 3 2 1n
R_2 2 5 31.47872218445133
R_4 0 3 1.699850997960372k
R_5 0 4 10k
R_6 4 5 2.137117626872152k
XOA 3 4 5 OAI
.ENDS CA

* Cella 2 (CB) *

.SUBCKT CB 1 5
C_11 2 1 68n
C_12 2 0 47n
C_3 3 2 0.1n
R_2 2 5 24.1793479621775
R_4 0 3 71.82673310105153k
R_5 0 4 10k
R_6 4 5 2.250676509173192k
XOA 3 4 5 OA
.ENDS CB

* Definizione circuito *

V1 1 0 AC 1
XCA 1 2 CA
XCB 2 3 CB

* Opzioni di analisi *

.AC DEC 101 9k 1.25MEG

.probe

.end

Dai risultati della simulazione, allegata con la presente relazione, si può osservare un comportamento anomalo
in banda passante: il fatto di aver introdotto una frequenza di taglio troppo bassa, ha accentuato il ripple in
banda passante.

Da un’analisi effettuata (i cui risultati verranno allegati alla relazione), si nota il fatto che, introducendo
una non idealità nella sola prima cella, provoca variazioni della funzione di trasferimento solo per frequenze

elevate, oltre a circa 50 kHz, mentre, in un intorno di 35 kHz, non si riscontrano sostanziali variazioni. Questo
comportamento è prevedibile, dal momento che la prima cella inizi ad attenuare solo da 68 kHz in poi, dunque
non possa influenzare in modo sensibile le frequenze più basse; idealizzando la sola prima cella, invece, si ha un
peggioramento del comportamento sia per quanto concerne l’inizio della banda passante, che le frequenze più
elevate; ciò è imputabile al fatto che la frequenza di taglio della seconda cella sia circa pari a 35 kHz. La forma
d’onda finale è una combinazione delle due, come osservato: in un intorno di 35 kHz sostanzialmente si ha il
solo contributo della prima cella, per frequenze più elevate entrambi i contributi.

3.2.2 Simulazione del comportamento in frequenza con ft = 300 kHz e aumento dei
componenti

Si ripete il discorso precedente, introducendo una frequenza di taglio dell’amplificatore operazionale pari a 300
kHz (un quinto della precedente), e aumentando i valori dei componenti della cella del 20 %. La simulazione
del comportamento in frequenza del circuito è stata dunque realizzata con il seguente script PSpice (contenente
i valori calcolati):

* Definizione amplificatore operazionale *

.SUBCKT OA 1 2 3
* | | |
* | | | output
* | inverting input
* non inverting input
G_1 0 4 1 2 1000
R_2 4 0 10k
C_3 4 0 530.516476973u
E_1 3 0 4 0 1
.ENDS OA

* Cella 1 (CA) *

.SUBCKT CA 1 5
C_11 2 1 120n
C_3 3 2 1.2n
R_2 2 5 37.77446662134159
R_4 0 3 2.039821197552446k
R_5 0 4 12k
R_6 4 5 2.564541152246582k
XOA 3 4 5 OA
.ENDS CA

* Cella 2 (CB) *

.SUBCKT CB 1 5
C_11 2 1 81.5999999999999n
C_12 2 0 56.4n
C_3 3 2 0.12n
R_2 2 5 29.0152175546130

R_4 0 3 86.192079721261836k
R_5 0 4 12k
R_6 4 5 2.7008118110078304k
XOA 3 4 5 OA
.ENDS CB

* Definizione circuito *

V1 1 0 AC 1
XCA 1 2 CA
XCB 2 3 CB

* Opzioni di analisi *

.AC DEC 101 9k 500k

.probe

.end

Si può osservare, come prevedibile, un ulteriore peggioramento rispetto al circuito di partenza: aumentando
i componenti del 20 % si esce violentemente dalle specifiche. Il fatto di ridurre ulteriormente la frequenza di
taglio degli amplificatori, inoltre, riduce a tal punto la banda passante del circuito, da non permettere neanche
ai ripple previsti dalla funzione di trasferimento originale di terminare.

